F04YAF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F04YAF returns elements of the estimated variance-covariance matrix of the sample regression coefficients for the solution of a linear least-squares problem.

The routine can be used to find the estimated variances of the sample regression coefficients.

2 Specification

```
 SUBROUTINE F04YAF(JOB, P, SIGMA, A, NRA, SVD, IRANK, SV, CJ, WORK,
1                  IFAIL)
 INTEGER           JOB, P, NRA, IRANK, IFAIL
 real              SIGMA, A(NRA,P), SV(P), CJ(P), WORK(P)
 LOGICAL           SVD
```

3 Description

The estimated variance-covariance matrix C of the sample regression coefficients is given by

$$C = \sigma^2 (X^T X)^{-1}, \quad X^T X \text{ non-singular,}$$

where $X^T X$ is the normal matrix for the linear least-squares regression problem

$$\min : \|y - Xb\|_2, \tag{1}$$

σ^2 is the estimated variance of the residual vector $r = y - Xb$, and X is an n by p observation matrix.

When $X^T X$ is singular, C is taken to be

$$C = \sigma^2 (X^T X)^{\dagger},$$

where $(X^T X)^{\dagger}$ is the pseudo-inverse of $X^T X$; this assumes that the minimal least-squares solution of (1) has been found.

The diagonal elements of C are the estimated variances of the sample regression coefficients, b.

The routine can be used to find either the diagonal elements of C, or the elements of the jth column of C, or the upper triangular part of C.

This routine must be preceded by a routine that returns either the upper triangular matrix U of the QU factorization of X or of the Cholesky factorization of $X^T X$, or the singular values and right singular vectors of X. In particular this routine can be preceded by one of the routines F04JAF, F04JDF or F04JGF, which return the parameters IRANK, SIGMA, A and SV in the required form. F04JGF returns the parameter SVD, but when this routine is used following routines F04JAF or F04JDF the parameter SVD should be set to .TRUE.. The parameter P of this routine corresponds to the parameter N of routines F04JAF and F04JGF.

4 References

[1] Anderson T W (1958) *An Introduction to Multivariate Statistical Analysis* Wiley

[2] Lawson C L and Hanson R J (1974) *Solving Least-squares Problems* Prentice–Hall

5 Parameters

1: JOB — INTEGER *Input*

On entry: JOB specifies which elements of C are required as follows:

JOB $= -1$	The upper triangular part of C is required.
JOB $= 0$	The diagonal elements of C are required.
JOB > 0	The elements of column JOB of C are required.

Constraint: $-1 \leq$ JOB $\leq$ P.

2: P — INTEGER *Input*

On entry: p, the order of the variance-covariance matrix C.

Constraint: P ≥ 1.

3: SIGMA — ***real*** *Input*

On entry: the standard error, σ, of the residual vector given by

$$\sigma = \sqrt{r^T r/(n-k)}, \quad n > k$$

$$\sigma = 0, \quad n = k,$$

where k is the rank of X.

Constraint: SIGMA ≥ 0.

4: A(NRA,P) — ***real*** array *Input/Output*

On entry: if SVD = .FALSE., A must contain the upper triangular matrix U of the QU factorization of X, or of the Cholesky factorization of $X^T X$; elements of the array below the diagonal need not be set.

If SVD = .TRUE., A must contain the first k rows of the matrix V^T, where k is the rank of X and V is the right-hand orthogonal matrix of the singular value decomposition of X. Thus the ith row must contain the ith right-hand singular vector of X.

On exit: if JOB ≥ 0, A is unchanged.

If JOB $= -1$, A contains the upper triangle of the symmetric matrix C. If SVD = .TRUE., elements of the array below the diagonal are used as workspace; if SVD = .FALSE., they are unchanged.

5: NRA — INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F04YAF is called.

Constraints:

if SVD = .FALSE., or JOB $= -1$, NRA $\geq$ P,
if SVD = .TRUE., and JOB ≥ 0, NRA $\geq$ max(1,IRANK).

6: SVD — LOGICAL *Input*

On entry: SVD must be .TRUE. if the least-squares solution was obtained from a singular value decomposition of X. SVD must be .FALSE. if the least-squares solution was obtained from either a QU factorization of X or a Cholesky factorization of $X^T X$. In the latter case the rank of X is assumed to be p and so is applicable only to full rank problems with $n \geq p$.

7: IRANK — INTEGER *Input*

On entry: if SVD = .TRUE., IRANK must specify the rank k of the matrix X. If SVD = .FALSE., IRANK is not referenced and the rank of X is assumed to be p.

Constraint: $0 <$ IRANK $\leq$ min(N,P).

8: SV(P) — ***real*** array *Input*

On entry: if SVD = .TRUE., SV must contain the first IRANK singular values of X. If SVD = .FALSE., SV is not referenced.

9: CJ(P) — ***real*** array *Output*

On exit: if JOB = 0, CJ returns the diagonal elements of C.

If JOB = $j > 0$, CJ returns the jth column of C.

If JOB = -1, CJ is not referenced.

10: WORK(P) — ***real*** array *Workspace*

If JOB > 0, WORK is not referenced.

11: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

On entry, P < 1,
or SIGMA < 0.0,
or JOB < -1,
or JOB $>$ P,
or SVD = .TRUE. and (IRANK < 0 or IRANK $>$ P)
or (JOB ≥ 0 and NRA $<$ max(1,IRANK))
or (JOB $= -1$ and NRA $<$ P)),
or SVD = .FALSE. and NRA$<$ P.

IFAIL = 2

On entry, SVD = .TRUE. and IRANK = 0.

IFAIL = 3

On entry, SVD = .FALSE. and overflow would occur in computing an element of C. The upper triangular matrix U must be very nearly singular.

IFAIL = 4

On entry, SVD = .TRUE. and one of the first IRANK singular values is zero. Either the first IRANK singular values or IRANK must be incorrect.

Overflow

If overflow occurs then either an element of C is very large, or more likely, either the rank, or the upper triangular matrix, or the singular values or vectors have been incorrectly supplied.

7 Accuracy

The computed elements of C will be the exact covariances of a closely neighbouring least-squares problem, so long as a numerically stable method has been used in the solution of the least-squares problem.

8 Further Comments

When JOB $= -1$ the time taken by the routine is approximately proportional to pk^2, where k is the rank of X. When JOB $= 0$ and SVD = .FALSE., the time taken by the routine is approximately proportional to pk^2, otherwise the time taken is approximately proportional to pk.

9 Example

To find the estimated variances of the sample regression coefficients (the diagonal elements of C) for the linear least-squares problem

$$\min r^T r, \text{ where } r = y - Xb \text{ and}$$

$$X = \begin{pmatrix} 0.6 & 1.2 & 3.9 \\ 5.0 & 4.0 & 2.5 \\ 1.0 & -4.0 & -5.5 \\ -1.0 & -2.0 & -6.5 \\ -4.2 & -8.4 & -4.8 \end{pmatrix}, \quad b = \begin{pmatrix} 3.0 \\ 4.0 \\ -1.0 \\ -5.0 \\ -1.0 \end{pmatrix},$$

following a solution obtained by F04JGF. See the routine document for further information.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F04YAF Example Program Text.
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          PMAX, NRX, LWORK
      PARAMETER        (PMAX=10,NRX=PMAX,LWORK=4*PMAX)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      real             SIGMA, TOL
      INTEGER          I, IFAIL, IRANK, J, JOB, N, P
      LOGICAL          SVD
*     .. Local Arrays ..
      real             CJ(PMAX), WORK(LWORK), X(NRX,PMAX), Y(PMAX)
*     .. External Subroutines ..
      EXTERNAL         F04JGF, F04YAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F04YAF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, P
      TOL = 5.0e-4
      IFAIL = 0
      IF (N.GT.0 .AND. N.LE.NRX .AND. P.GT.0 .AND. P.LE.PMAX) THEN
         READ (NIN,*) ((X(I,J),J=1,P),I=1,N)
         READ (NIN,*) (Y(I),I=1,N)
*
         CALL F04JGF(N,P,X,NRX,Y,TOL,SVD,SIGMA,IRANK,WORK,LWORK,IFAIL)
*
         WRITE (NOUT,*)
         WRITE (NOUT,99999) 'SIGMA =', SIGMA, '  Rank =', IRANK,
     +     '  SVD =', SVD
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Solution vector'
```

```
         WRITE (NOUT,99998) (Y(I),I=1,P)
         JOB = 0
*
         CALL F04YAF(JOB,P,SIGMA,X,NRX,SVD,IRANK,WORK,CJ,WORK(P+1),
     +               IFAIL)
*
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Estimated variances of regression coefficients'
         WRITE (NOUT,99998) (CJ(J),J=1,P)
      END IF
      STOP
*
99999 FORMAT (1X,A,F9.4,A,I3,A,L3)
99998 FORMAT (1X,7F9.4)
      END
```

9.2 Program Data

```
F04YAF Example Program Data
   5   3
 0.60  1.20  3.90
 5.00  4.00  2.50
 1.00 -4.00 -5.50
-1.00 -2.00 -6.50
-4.20 -8.40 -4.80
 3.0   4.0  -1.0  -5.0  -1.0
```

9.3 Program Results

```
 F04YAF Example Program Results

 SIGMA =   0.4123  Rank =  3  SVD =  F

 Solution vector
    0.9533  -0.8433   0.9067

 Estimated variances of regression coefficients
    0.0106   0.0093   0.0045
```